

# Manual de las funciones financieras del Excel

Jairo Gutiérrez Carmona  
gutierrezcarmona@gmail.com

Marzo de 2010

---

## Contenido

<b>INTRODUCCIÓN</b>	<b>2</b>
<b>1 ASISTENTE PARA FUNCIONES</b>	<b>4</b>
<b>2 MANUAL DE LAS FUNCIONES FINANCIERAS DEL EXCEL</b>	<b>7</b>
<b>2.1 FUNCIONES PARA CONVERSION DE TASAS</b>	<b>8</b>
2.1.1 INT.EFECTIVO	8
2.1.2 TASA.NOMINAL	9
<b>2.2 FUNCIONES DE PAGO UNICO Y SERIES UNIFORMES</b>	<b>12</b>
2.2.1 VF	15
2.2.2 VA	15
2.2.3 PAGO	16
2.2.4 TASA	17
2.2.5 NPER	17
2.2.6 PAGOINT	18
2.2.7 PAGOPRIN	20
2.2.8 PAGO.INT.ENTRE	20
2.2.9 PAGO.PRINC.ENTRE	21
<b>2.3 FUNCIONES DE EVALUACION DE PROYECTOS</b>	<b>23</b>
2.3.1 VNA	23
2.3.2 TIR	25
2.3.3 TIRM	26
2.3.4 VNA.NO.PER	27
2.3.5 TIR.NO.PER	28
<b>2.4 FUNCIONES DE INVERSIONES FINANCIERAS</b>	<b>29</b>
2.4.1 TASA.INT	30

2.4.2	INT.ACUM.V	30
2.4.3	INT.ACUM	31
2.4.4	PRECIO.DESCUENTO	31
2.4.5	RENDTO.DESC	32
2.4.6	CANTIDAD.RECIBIDA	33
2.4.7	TASA.DESC	34
2.4.8	CUPON.NUM	35
2.4.9	CUPON.DIAS	35
2.4.10	CUPON.DIAS.L1	35
2.4.11	CUPON.DIAS.L2	36
2.4.12	CUPON.FECHA.L1	36
2.4.13	CUPON.FECHA.L2	36
2.4.14	VF.PLAN	37
2.4.15	PRECIO.VENCIMIENTO	37
2.4.16	RENDTO.VENCTO	39
2.4.17	PRECIO	39
2.4.18	RENDTO	40
2.5	FUNCIONES PARA MEDIR EL RIESGO DE LAS INVERSIONES	41
2.5.1	DURACION	41
3	BIBLIOGRAFÍA	45
4	ANEXO N. 1	46

---

## INTRODUCCIÓN

Este documento está orientado a la utilización de las Funciones Financieras que vienen con la hoja de cálculo Excel.

No obstante la versatilidad de las hojas de cálculo, construir las formulas cada vez que se soluciona un problema financiero requiere un gran esfuerzo; afortunadamente el Excel se complementa con funciones que son formulas programadas para resolver problemas estándar de una manera fácil, pero principalmente rápida. Por lo tanto, conocer la operación real de estas funciones y en especial el significado de sus respuestas, es un gran paso en la solución veloz de problemas financieros para mejorar el análisis de las respuestas.

El documento se divide en dos partes:

- **Asistente para funciones**, para las personas que se inician en la utilización de las funciones, es una guía sobre cómo utilizar este asistente que facilita la forma de insertar funciones en una hoja de cálculo. Quienes ya han utilizado la herramienta pueden omitir esta parte.
- **Manual de las funciones financieras del Excel**, es una explicación sobre la forma de operación de funciones financieras que vienen con la hoja de cálculo Excel, agrupándolas según el problema que solucionan.

Finalmente se incluye un anexo con las funciones tratadas en este fascículo, pero ordenadas alfabéticamente para facilitar su ubicación en el texto.

## 1 ASISTENTE PARA FUNCIONES

Las funciones son fórmulas que vienen programadas en el Excel y resuelven problemas generales; su característica principal es que de una manera rápida y sencilla ejecutan cálculos complejos.

Para realizar los cálculos las funciones utilizan valores denominados argumentos, que deben introducirse en un orden determinado y dejan el resultado que calculan en la celda donde han sido escritas.

Todas las funciones del Excel constan de tres partes:

1. El signo igual (=) que precede toda función para indicar que es una fórmula (cuando la función está empleada como argumento, es decir que no aparece al principio de la entrada, no requiere el signo igual),
2. El nombre de la función, que tiene una ortografía definida y por lo tanto siempre debe escribirse de la misma manera,
3. Los argumentos que tienen una sintaxis definida y por lo tanto siempre se deben escribir en el mismo orden. Los argumentos van entre paréntesis y hacen referencia a los datos, celdas o rangos sobre los cuales operará la función. Los argumentos pueden ser números, direcciones de celda, nombres de celda u otras funciones.

El siguiente es el formato general de una función:

**= NombreFunción(argumento\_1,argumento\_2, .... ,argumento\_n)**


El principio de las funciones es ahorrar tiempo en los cálculos dispendiosos, por lo tanto más que saber su mecánica de funcionamiento, es importante tener los conocimientos teóricos de finanzas y matemáticas financieras para plantear los problemas e interpretar los resultados que arroja.

Para facilitar este trabajo el Excel cuenta con un Asistente para Funciones que guía al usuario en su utilización, ya que ofrece ayuda con una breve descripción de la función y un cuadro de diálogo con los argumentos que se deben diligenciar para no cometer errores de ortografía, de sintaxis o teóricos, pues también presenta breves descripciones de los argumentos utilizados.


Para tener acceso al asistente para funciones se puede proceder de varias maneras:

- ✓ En Excel 2003 se tienen tres opciones:
  - Seleccionar **Insertar** | **Función** en el menú de texto

- Seleccionar el icono Insertar Función de la barra estándar



- Oprimir el icono del Asistente para Funciones de la barra de fórmulas



- ✓ En Excel 2007 se tienen dos opciones:

- Seleccionar la ficha FORMULAS y en el grupo de tareas BIBLIOTECA DE FUNCIONES se encuentra el icono de las funciones financieras




- Oprimir el icono del Asistente para Funciones de la barra de fórmulas, igual que en el Excel 2003


- ✓ Además siempre existe la posibilidad de introducir manualmente las funciones y el Excel ofrece una ayuda sobre la sintaxis (no hay descripción del uso de la función)

Cualquiera de los sistemas que se utilice deja a disposición del usuario el Asistente para Funciones, que es una ventana con cuatro áreas: una para la categoría de funciones dentro de la cual se encuentra la categoría financieras; otra para seleccionar la función deseada; debajo de ellas se encuentra el área de ayuda donde se aprecian los argumentos de cada función y un corto resumen de los resultados que se obtienen con ella; por último el área de botones, en la cual una vez seleccionada la función a ejecutar se oprime el botón Aceptar y se despliega el segundo paso del Asistente.

El siguiente es el aspecto del paso uno del Asistente para Funciones:



El siguiente es el aspecto del paso dos del Asistente para Funciones



## 2 MANUAL DE LAS FUNCIONES FINANCIERAS DEL EXCEL

Para efecto de desarrollar este fascículo, las funciones financieras se han clasificado en cinco categorías, de acuerdo con el uso que se dé a los resultados que producen. Tales categorías son:

1. Funciones que sirven para convertir tasas
2. Funciones para resolver problemas de pago único y series uniformes
3. Funciones para facilitar la evaluación de proyectos
4. Funciones especializadas en el análisis de inversiones financieras
5. Funciones para medir el riesgo de las inversiones

Al iniciar la presentación de cada categoría de funciones se exponen y comentan los argumentos que utilizan, con el fin de facilitar el posterior entendimiento de la forma de operación de la función y sus ejemplos.

A continuación se hace una presentación de cada una de las funciones según su clasificación, dicha presentación incluye la definición de la función, la sintaxis que debe tener<sup>1</sup>, las observaciones sobre los argumentos que se utilizan o sobre la interpretación de los resultados que arroja la función, y en los casos en que se considere necesario se presentarán ejemplos de su utilización.

---

<sup>1</sup> La sintaxis se refiere al orden en que se deben incluir los argumentos de la función. En caso de introducir la función manualmente es indispensable conocer y recordar esta sintaxis; si se utiliza el asistente para funciones la sintaxis se organiza automáticamente, sin importar el orden en que se incluyan los argumentos.

## 2.1 FUNCIONES PARA CONVERSION DE TASAS

Dentro de este grupo se clasifican dos funciones que sirven para convertir tasas de interés efectivas en nominales y viceversa. Los argumentos que utilizan las funciones financieras para conversión de tasas son los siguientes:

ARGUMENTO	SIGNIFICADO Y OBSERVACIONES
Núm_per y Núm_per_año	Es la cantidad de períodos de capitalización de interés que hay en un año. Por período de capitalización se entiende el tiempo que transcurre entre dos fechas de liquidación de interés, pero en el caso de estas funciones se supone que el interés pagado no se retira ni se consume, si no que es reinvertido por el tiempo que resta del año.
Int_nominal	Es la tasa de interés nominal anual expresada en términos decimales. Se debe tener en cuenta que es la tasa anual y no la del período de capitalización.
Tasa_efectiva	Es la tasa de interés efectiva anual, es decir la rentabilidad que efectivamente se recibiría si los intereses que se perciben por la inversión son reinvertidos en las mismas condiciones por el tiempo que resta del año.

### 2.1.1 INT.EFECTIVO

Devuelve la tasa de interés efectiva anual, si se conocen la tasa de interés nominal anual y el número de períodos de capitalización de interés que hay en un año.

#### INT.EFECTIVO(int\_nominal, núm\_per\_año)

	A	B	C
1			
2	<b>FUNCION INT.EFECTIVO</b>		
3			
4	TASA NOMINAL ANUAL	20.00%	
5	NUMERO DE PERIODOS	4.00	
6			
7	TASA EFECTIVA ANUAL		
8	CON FUNCION	21.55%	=INT.EFECTIVO(B4;B5)
9			
10	CON FORMULA	21.55%	=(1+B4/B5)^B5-1
11			
12			
13	<b>El resultado con la función es igual al</b>		
14	<b>resultado con la formula solo en el caso</b>		
15	<b>en que el número de periodos sea un entero</b>		

Hay que tener especial cuidado con esta función, ya que solo produce resultados confiables cuando la cantidad de períodos de pago en el año (núm\_per\_año) tiene

valores exactos, por ejemplo mensual (12), trimestral (4), semestral (2) o anual (1). Cuando los períodos son irregulares este argumento se trunca a entero y el resultado no es real, tal como se aprecia a continuación:

	A	B	C	D	E	F	G	H
1								
2	<b>FUNCION INT.EFECTIVO</b>							
3								
4	FECHA INICIO	15-abr-09						
5	FECHA FINAL	30-jun-09						
6	DURACION	75 dias						
7	VALOR INICIAL	\$ 1,000,000						
8	VALOR FINAL	\$ 1,066,667						
9								
10	PERIODOS EN EL AÑO	4.80	<== VALOR NO EXACTO					
11			(4.8 PAGOS AL AÑO ==> CADA DOS MESES Y MEDIO)					
12	RENDIMIENTO NOMINAL	32.00%						
13								
14	RENDIMIENTO EFECTIVO	36.31%	<== CALCULADO CON FORMULA					
15			=(1+B10/B8)^B8-1					
16								
17	RENDIMIENTO EFECTIVO	36.05%	<== CALCULADO CON FUNCION					
18			=INT.EFECTIVO(B12;B10)					

Los períodos de pago no son exactos y por lo tanto el resultado no es real

### 2.1.2 TASA.NOMINAL

Devuelve la tasa de interés nominal anual, si se conocen la tasa de interés efectiva anual y el número de períodos de capitalización de interés que hay en un año.

**TASA.NOMINAL(tasa\_efectiva, núm\_per)**

### OBSERVACIONES A LAS FUNCIONES PARA CONVERSION DE TASAS

El argumento núm\_per\_año se trunca a entero.

Si alguno de los argumentos int\_nominal o tasa\_efectiva es menor o igual a cero o si el argumento núm\_per o núm\_per\_año es menor que uno, la función devuelve el valor de error #¡NUM!

En ambas funciones la respuesta que se obtenga se expresa en términos decimales y debe dársele el formato de porcentaje. Nunca divida ni multiplique por cien el resultado de estas funciones, ya que no podrá utilizarlo como porcentaje en cálculos posteriores.

Estas funciones se limitan a calcular el interés efectivo o nominal en el caso del pago de intereses vencidos. Si el problema se refiere a intereses anticipados debe encontrar la respuesta mediante la fórmula. Para su información se incluyen a continuación las fórmulas para calcular el interés efectivo y nominal anual en las modalidades de pago vencidas y anticipadas:

Fórmula para el interés **efectivo** en modalidad **vencida**:

$$EA = \left(1 + \frac{NA}{n}\right)^n - 1$$

Fórmula para el interés **nominal** en modalidad **vencida**:

$$NA = \left[ \left(1 + EA\right)^{\frac{1}{n}} - 1 \right] * n$$

Fórmula para el interés **efectivo** en modalidad **anticipada**:

$$EA = \frac{1}{\left(1 - \frac{NA}{n}\right)^n} - 1$$

Fórmula para el interés **nominal** en modalidad **anticipada**:

$$NA = \left(1 - \frac{1}{\left(1 + EA\right)^{\frac{1}{n}}}\right) * n$$

Donde:

NA es el interés nominal anual

EA es el interés efectivo anual

n es la cantidad de períodos de capitalización de interés que hay en un año

## EQUIVALENCIA DE TASAS CON FUNCIONES FINANCIERAS

Por definición dos tasas de interés son equivalentes cuando arrojan la misma tasa efectiva y tradicionalmente en la conversión manual de tasas de interés se utiliza la siguiente fórmula:

$$\left(1 + \frac{NA_1}{n}\right)^n = EA + 1 = \left(1 + \frac{NA_2}{m}\right)^m$$

Donde:

$NA_1$  es una tasa de interés nominal conocida, que liquida interés n veces

$NA_2$  es una tasa de interés nominal desconocida, que liquida intereses m veces

$EA$  es la tasa de interés efectiva común para  $NA_1$  y para  $NA_2$

Dadas las características de las dos funciones para conversión de tasas que vienen con el Excel, solo es posible calcular equivalencias de tasas de interés para la modalidad vencida.

Para calcular equivalencias en la hoja de cálculo Excel se requiere anidar dos funciones<sup>2</sup>: calcular la tasa nominal desconocida utilizando como uno de los parámetros la tasa efectiva, calculada a su vez a partir de la tasa nominal conocida.

Quiere decir que funciona con la misma filosofía de todas las equivalencia: nominal conocida → efectiva → nominal desconocida. A continuación se presenta un ejemplo:

	A	B	C	D	E	F
1						
2	<b>EQUIVALENCIAS DE TASAS</b>					
3						
4	TASA NOMINAL CONOCIDA	12.00%	nominal anual	mes	vencido	
5	PERIODICIDAD DE PAGO	12	trimestral			
6						
7	PERIODICIDAD DE PAGO	4	mensual			
8	TASA NOMINAL DESCONOCIDA	12.12%	nominal anual	trimestre	vencido	


<sup>2</sup> Las funciones anidadas usan una función como uno de los argumentos de otra. Puede anidar hasta 64 niveles de funciones.

## 2.2 FUNCIONES DE PAGO UNICO Y SERIES UNIFORMES


En Excel existe una serie de funciones que sirven para resolver las dos clases de problemas, por lo tanto es importante definir en qué consiste cada uno:

- Los problemas de pago único son aquellos en los cuales **durante el desarrollo del negocio no existen movimientos de dinero**. Se parte de un valor inicial ( $P$ ) para llegar a un valor final ( $F$ ), tiempo durante el cual los intereses se causan y se capitalizan.

Para el caso del pago único solo existen problemas de capitalización con pago vencido.



- Los problemas de series uniformes son aquellos en los cuales **entre el inicio y el final del negocio existen movimientos de dinero** por concepto de cuotas. En estos problemas se supone que los valores recibidos o pagados durante el tiempo del negocio son reinvertidos por lo que resta del plazo total, en las mismas condiciones existentes para la inversión original.




Un problema se considera de series uniformes cuando reúne todas las siguientes condiciones:

- El monto de los pagos es constante
- La periodicidad de los pagos es constante
- La tasa de interés es constante
- Existen  $n$  cuotas en la serie

Para el caso de series uniformes existen dos clases de problemas: los de amortización o pago de deudas y los de capitalización o de ahorro; ambos pueden ser con pagos vencidos o anticipados. A continuación se ilustra, a través del flujo de caja, las dos clases de problemas:


- ✓ **Problemas de amortización:** se refieren al pago de deudas y se caracterizan por iniciar el negocio con un desembolso que se debe amortizar durante un plazo con una serie de pagos periódicos iguales.



Un valor presente, equivalente a varios pagos futuros iguales

Las variables que intervienen son el valor inicial ( $P$ ), los pagos periódicos o anualidades ( $A$ ), el plazo ( $n$ ), la tasa de interés periódica ( $i$ ) y en algunas ocasiones un pago final ( $F$ ), tal como se muestra en el gráfico siguiente:

✓ **Problemas de capitalización:** se refieren a la constitución de un ahorro; el negocio comienza sin un saldo inicial y durante un determinado plazo se ahorran cuotas periódicas iguales hasta reunir un valor final. Las variables utilizadas son las mismas que en el caso anterior.



Una serie de pagos iguales,  
equivalente a un monto futuro

Atendiendo a las variables que se utilizan en estos problemas, a continuación se presentan las convenciones que se utilizarán en la exposición de las funciones financieras:

- P:** Desembolso que se hace al iniciar el negocio (en el momento cero)
- A:** Pagos iguales efectuados a intervalos iguales (anualidades)
- n:** Cantidad de pagos iguales (no necesariamente igual al plazo del negocio)
- i:** Tasa periódica de interés del negocio (costo o rentabilidad)
- F:** Desembolso que se hace al final del negocio o monto que se desea alcanzar con los ahorros (en el momento  $n$ )

En las soluciones de estos problemas siempre hay equivalencia entre los pagos periódicos iguales y el valor presente o futuro que se halle, por lo tanto la variable  $P$  normalmente se refiere al desembolso inicial pero en algunos casos puede asimilarse al valor presente de los pagos que se hagan durante la vida del negocio. Igual ocurre con la variable  $F$  que se refiere al desembolso final pero también puede asimilarse al valor futuro de los mismos pagos.

Tanto para los problemas de pago único como para los de series uniformes se utilizan las mismas funciones financieras del Excel; en el caso de los problemas de series uniformes dichas funciones dan respuesta a los cinco términos mencionados; en tanto que para los de pago único solo dan respuesta a cuatro de los términos, ya que, como no existen pagos durante la vida del negocio, no se utiliza la variable  $A$  (anualidades).

Los argumentos que utilizan las funciones financieras de pago único y series uniformes son los siguientes:

ARGUMENTO	SIGNIFICADO Y OBSERVACIONES						
Va	P en los términos financieros. Es el desembolso inicial o el valor actual de una serie de pagos futuros iguales. Si este argumento se omite, se considerará 0.						
Pago	A en los términos financieros. Es el pago que se efectúa en cada período y que no cambia durante la vida de la anualidad. El Pago incluye el capital y el interés pero no incluye ningún otro cargo o impuesto. Este argumento debe tener signo contrario al de Va, para conservar las condiciones del flujo de caja: los ingresos se presentan con signo positivo y los egresos con signo negativo.						
Nper	n en los términos financieros. Es la cantidad total de períodos en una anualidad.						
Tasa	i en los términos financieros. Es la tasa de interés por período. Tener en cuenta que no es la tasa anual, si no la tasa del período de pago expresada en términos decimales. Es importante mantener la uniformidad en el uso de las unidades con las que se especifica Tasa y Nper.						
Vf	F en los términos financieros. Es el valor futuro o el saldo en efectivo que desea lograr después de efectuar el último pago. Si el argumento Vf se omite, se asume que el valor es 0.						
Tipo	Es el número 0 ó 1 e indica la forma de pago de las cuotas entre vencido y anticipado <table style="width: 100%; border: none;"> <tr> <td style="text-align: center;"><u>Defina tipo como</u></td> <td style="text-align: center;"><u>Si el pago vence</u></td> </tr> <tr> <td style="text-align: center;">0 ó se omite</td> <td style="text-align: center;">Al final del período</td> </tr> <tr> <td style="text-align: center;">1</td> <td style="text-align: center;">Al principio del período</td> </tr> </table> <p>Como en los problemas de pago único no existen pagos, siempre se debe omitir este argumento.</p>	<u>Defina tipo como</u>	<u>Si el pago vence</u>	0 ó se omite	Al final del período	1	Al principio del período
<u>Defina tipo como</u>	<u>Si el pago vence</u>						
0 ó se omite	Al final del período						
1	Al principio del período						
Período	Especifica el número ordinal de la cuota que se está estudiando, que debe encontrarse en el intervalo comprendido entre 1 y Nper.						
Per_inicial y Per_final	Especifican el número ordinal de la primera y la última cuota de un lapso de tiempo al cual se le van a analizar las cuotas pagadas.						
Estimar	Es una tasa de interés estimada para que el Excel empiece las iteraciones en el cálculo de la tasa de interés de una serie uniforme. Si el argumento Estimar se omite, se supone que es 10%.						

### 2.2.1 VF

Devuelve el valor futuro de una inversión, equivalente a los pagos constantes que se hacen periódicamente y a una tasa de interés constante.

#### VF(tasa, nper, pago, va, tipo)

Por ejemplo, si uno deposita \$ 250.000 mensuales durante 2 años en una cuenta que paga el 36% nominal anual desea saber cuanto dinero tendrá ahorrado al final, la respuesta la puede obtener con la función VF, como se aprecia en el siguiente ejemplo:

	A	B	C	D	E
1	TASA	36%			
2	NPER	24			
3	PAGO	250,000			
4	TIPO	0			
5					
6	VF	8,606,618			
7					
8	VF				
9	Tasa	B1/12	=	0.03	
10	Nper	B2	=	24	
11	Pago	-B3	=	-250000	
12	Va		=	número	
13	Tipo	B4	=	0	
14					= 8606617.554

Hay tres aspectos que se deben tener cuenta en este ejemplo:

- El interés que se incluye en el argumento Tasa debe estar en la misma unidad de tiempo que se use para el argumento Nper, en este caso, como son cuotas mensuales, la tasa de interés debe ser mensual, por lo tanto hay que dividir por doce la tasa anual nominal (B1/12).
- Va puede omitirse como se aprecia en el asistente para funciones y en la barra de fórmulas automáticamente deja el espacio en la función, asumiéndolo como cero.
- Si se desea que las cifras en la hoja de cálculo sean positivas, el argumento Pago debe introducirse con signo negativo, como se aprecia en el asistente para funciones (-B3)

### 2.2.2 VA

Devuelve el valor actual de una inversión que recibirá abonos periódicos constantes en el futuro, a una tasa de interés constante. Cuando pide dinero prestado, el monto del préstamo es el valor actual para el prestamista, que es equivalente a las cuotas iguales que recibirá en el futuro. Esta función conserva las misma observaciones efectuadas para VF.

#### VA(tasa, nper, pago, vf, tipo)

Por ejemplo, cuál será el valor de un préstamo que se paga en año y medio con cuotas mensuales iguales de \$2.500.000, si la tasa de interés es del 36% nominal anual; la respuesta se obtiene con la función VA, como se muestra a continuación:

	A	B	C	D	E
1	TASA	36%			
2	NPER	18			
3	PAGO	2,500,000			
4	TIPO	0			
5					
6	VA	34,383,783			=VA(B1/12,B2,-B3,,B4)
7	VA				
8		Tasa	B1/12	=	0.03
9		Nper	B2	=	18
10		Pago	-B3	=	-2500000
11		Vf		=	número
12		Tipo	B4	=	0
13					
14					= 34383782.7

### 2.2.3 PAGO

Devuelve el pago periódico de una anualidad basándose en pagos constantes y en una tasa de interés constante. La respuesta que arroja esta función es equivalente a Va y a Vf, dadas las condiciones que se plantean en los períodos, la tasa y la cuota.

#### PAGO(tasa, nper, va, vf, tipo)

Por ejemplo, si se recibe un crédito de \$5.000.000 para ser cancelado en 20 cuotas trimestrales iguales (cinco años), a una tasa nominal anual del 18% trimestre vencido, el valor de cada cuota se puede calcular con la función Pago. Aquí se omite Vf y en la función es reemplazado por cero.

	A	B	C	D	E
1	TASA	18%			
2	NPER	20 trimestres			
3	VA	5,000,000			
4	Vf	0			
5	TIPO	0			
6					
7	PAGO	384,381			=PAGO(B1/4,B2,-B3,,B5)
8	PAGO				
9		Tasa	B1/4	=	0.045
10		Nper	B2	=	20
11		Va	-B3	=	-5000000
12		Vf		=	número
13		Tipo	B5	=	0
14					
15					= 384380.7216

En algunos problemas puede darse la necesidad de requerir tanto VA como VF para poder resolver el problema, como es por ejemplo el caso del leasing, en el cual, además del valor inicial de un equipo se presentan unas cuotas mensuales iguales y al final el pago de la una opción de compra para que el usuario adquiera el equipo: en un leasing de \$ 80.000.000 a 36 meses con una

	A	B	C	D	E
1	TASA	1.30%			
2	NPER	36			
3	VA	80,000,000			
4	OPCION	10%			
5	TIPO	0			
6					
7	PAGO	2,621,111			=PAGO(B1,B2,-B3,B4*B3,B5)
8	PAGO				
9		Tasa	B1	=	0.013
10		Nper	B2	=	36
11		Va	-B3	=	-80000000
12		Vf	B4*B3	=	8000000
13		Tipo	B5	=	0
14					
15					= 2621110.75

tasa de interés del 1.3% mensual y una opción de compra del 10%, la función Pago para calcular la cuota a pagar funcionaría así:

Vale hacer dos observaciones:

- Como el negocio se está viendo desde el punto de vista del prestamista, el desembolso o sea Va se presenta negativo, para que los ingresos (Pago) aparezcan positivos, por lo tanto Vf debe introducirse positivo en la función
- Además, el argumento Vf se introduce como una operación, multiplicando el porcentaje de la opción (B4) por el valor del equipo (B3)

### 2.2.4 TASA

Devuelve la tasa de interés por período de pago de una anualidad; esta tasa es la que hace equivalentes el pago constante con el valor presente o con el valor futuro.

#### TASA(nper, pago, va, vf, tipo, estimar)

En las series uniformes Tasa se calcula por iteración y puede no tener solución. El argumento estimar es opcional y en caso de omitirse las iteraciones inician en 10%.

Por ejemplo a que tasa de interés se concede un préstamo de \$30.000.000 que paga en 36 cuotas mensuales iguales de \$1.300.000:

	A	B	C	D	E
1	NPER		36		
2	VA		30,000,000		
3	PAGO		1,300,000		
4	TIPO		0		
5					
6	TASA		2.63%		=TASA(B1,B3,-B2,,B4)
7	TASA				
8	Nper	B1	=	36	
9	Pago	B3	=	1300000	
10	Va	-B2	=	-30000000	
11	Vf		=	número	
12	Tipo	B4	=	0	
13					
14					= 0.026334347

se

Como el argumento Nper está expresado en meses, la respuesta se presenta también en meses.

### 2.2.5 NPER

Devuelve la cantidad de períodos que debe tenerse una inversión para que sea equivalente a una serie de pagos periódicos iguales.

#### NPER(tasa, pago, va, vf, tipo)

La unidad de tiempo en la que se expresa Nper es la misma que se emplee para la tasa interés.

Por ejemplo, durante cuanto tiempo se deben hacer depósitos mensuales iguales de \$200.000, para reunir un saldo total de \$20.000.000, si se invierten en una cuenta que paga el 0.5% mensual:

	A	B	C	D	E															
			=NPER(B1,-B3,,B2,B4)																	
1	TASA	0.41%																		
2	VF	20,000,000																		
3	PAGO	200,000																		
4	TIPO	0																		
5																				
6	<b>NPER</b>	<b>84</b>	=NPER(B1,-B3,,B2,B4)																	
7	<table border="1"> <tr> <td>Tasa</td> <td>B1</td> <td>= 0.0041</td> </tr> <tr> <td>Pago</td> <td>-B3</td> <td>= -200000</td> </tr> <tr> <td>Va</td> <td></td> <td>= número</td> </tr> <tr> <td>Vf</td> <td>B2</td> <td>= 20000000</td> </tr> <tr> <td>Tipo</td> <td>B4</td> <td>= 0</td> </tr> </table>					Tasa	B1	= 0.0041	Pago	-B3	= -200000	Va		= número	Vf	B2	= 20000000	Tipo	B4	= 0
Tasa	B1	= 0.0041																		
Pago	-B3	= -200000																		
Va		= número																		
Vf	B2	= 20000000																		
Tipo	B4	= 0																		
8																				
9																				
10																				
11																				
12																				
13																				
14					= 83.97404462															

### 2.2.6 PAGOINT

Devuelve el monto del interés pagado por una inversión o un crédito en un período determinado, basándose como ya se ha dicho en pagos periódicos constantes y en una tasa de interés constante.

#### **PAGOINT(tasa, período, nper, va, vf, tipo)**

A los argumentos que se han venido manejando hasta ahora se agrega el argumento **Período** que representa el número ordinal de la cuota que se está analizando. Recuérdese que en las series uniformes el valor futuro y el valor presente son equivalentes a una serie de pagos periódicos iguales que se realizan durante un tiempo determinado y que estos pagos incluyen dos componentes:

- El interés sobre el saldo de la inversión o del crédito
- El abono a capital de la inversión o del crédito

El primer componente se calcula con la función Pagoint (pago de interés) que se explica en este numeral y el segundo componente con la función Pagoprin (pago del principal o capital) que se explica en el siguiente numeral.

	A	B	C	D	E	F	G
1	CREDITO	\$ 5,000,000	PLAZO	4	trimestres		
3	INTERES	32% TV					
5	CUOTA No.	SALDO INICIAL	VALOR CUOTA	ABONO CAPITAL	PAGO DE INTERES	SALDO FINAL	
7	1	5,000,000	1,509,604	1,109,604	400,000	3,890,396	
8	2	3,890,396	1,509,604	1,198,372	311,232	2,692,024	
9	3	2,692,024	1,509,604	1,294,242	215,362	1,397,782	
10	4	1,397,782	1,509,604	1,397,782	111,823	0	
11							
12		=PAGO((\$B\$3/\$E\$1), \$E\$1, B1)				Calcula el monto de la cuota	
13							
14		=PAGOPRIN((\$B\$3/\$E\$1), A7, \$E\$1, -\$B\$1)				Calcula el abono a capital	
15							
16		=PAGOINT((\$B\$3/\$E\$1), A7, \$E\$1, -\$B\$1)				Calcula el pago de intereses	

Es importante tener en cuenta que de acuerdo con las costumbres del sector financiero colombiano, las funciones Pagoint y Pagoprin solo arrojan respuestas aceptables en el caso de la modalidad de pagos vencidos, ya que para los pagos de cuota anticipada se presenta una discordancia en la forma de contabilización de los intereses.

Por ejemplo, si se pretende elaborar la tabla de liquidación de un crédito de \$5.000.000 concedido el 15 de mayo al 32% nominal anual, pagadero en un año en cuotas trimestrales anticipadas, al resolver el ejercicio con formulas, se tiene el siguiente resultado:

	A	B	C	D	E	F	G
1	VA	5,000,000					
2	TASA	32%					
3	NPER	4 trimestres	PAGO ==>	1,397,782			
4					=PAGO(B2/4, B3, -B1, 1)		
5							
6	PERIODO	FECHA	SALDO INICIAL	VALOR CUOTA	INTERES	ABONO A CAPITAL	SALDO FINAL
7	0	15-may	5,000,000	1,397,782	288,177	1,109,604	3,890,396
8	1	15-ago	3,890,396	1,397,782	199,409	1,198,372	2,692,024
9	2	15-nov	2,692,024	1,397,782	103,539	1,294,242	1,397,782
10	3	15-feb	1,397,782	1,397,782	0	1,397,782	0

Como se aprecia, el valor de los intereses pagados en la primera cuota se calcula multiplicando el valor del dinero ajeno utilizado (C7 – D7) por la tasa de interés periódica. La función Pagoprin no cumple con este concepto, por lo tanto no es

conveniente utilizarla en el caso de cuotas anticipadas<sup>3</sup>, como se observa a continuación:

E20	=PAGOINT(\$B\$15/4,A20,\$B\$16,-\$B\$14,,1)						
	A	B	C	D	E	F	G
14	VA	5,000,000					
15	TASA	32%					
16	NPER	4 trimestres	PAGO ==>		1,397,782		
17					=PAGO(B2/4,B3,-B1,,1)		
18							
19	PERIODO	FECHA	SALDO INICIAL	VALOR CUOTA	INTERES	ABONO A CAPITAL	SALDO FINAL
20	1	15-may	5,000,000	1,397,782	0	1,397,782	3,602,218
21	2	15-ago	3,602,218	1,397,782	288,177	1,109,604	2,492,614
22	3	15-nov	2,492,614	1,397,782	199,409	1,198,372	1,294,242
23	4	15-feb	1,294,242	1,397,782	103,539	1,294,242	0

Cuando la tabla se calcula con funciones en el primer mes no se cobran intereses, cuando en realidad el prestatario utilizó durante un mes los \$5.000.000 del crédito menos la cuota de \$1.397.782 que canceló como primera cuota anticipada, es decir que debe pagar intereses sobre \$3.602.218 que al 8% arrojan un total de \$288.177. Como puede apreciarse en los dos cuadros anteriores, el valor de los intereses es igual pero tienen un período de desfase, por lo tanto, al calcular la verdadera rentabilidad de la operación se presentan diferencias, sin contar con las inconsistencias contables que se generan.

### 2.2.7 PAGOPRIN

Devuelve el monto abonado al capital de una inversión o de un crédito en un período determinado, basándose como ya se ha dicho en pagos periódicos constantes y en una tasa de interés constante.

**PAGOPRIN(tasa, período, nper, va, vf, tipo)**

### 2.2.8 PAGO.INT.ENTRE

Devuelve el monto de los intereses pagados entre dos fechas, para lo cual basta con dar los datos del crédito y el número ordinal de las cuotas a pagar donde empieza y termina el período.

<sup>3</sup> Hay que tener en cuenta que no es lo mismo cuotas anticipadas que intereses anticipados.

**PAGO.INT.ENTRE(tasa, nper, vp, per\_inicial, per\_final, tipo)**

Obviamente, se tiene como condición que las cuotas que se paguen por el crédito sean constantes. Por ejemplo si se desea saber cuál es el monto de los intereses pagados durante los dos primeros años de un crédito de vivienda por \$50.000.000, adquirido a 180 meses a una tasa del 3%, se deben dar los siguientes argumentos a la función:

A3		=	=-PAGO.INT.ENTRE(3%,180,50000000,1,24,0)
	A	B	C
1			
2			
3	\$	35,923,148	
4			
5	PAGO.INT.ENTRE(tasa, nper, vp, per_inicial, per_final, tipo)		
6			

Esta función no acepta signo negativo en el argumento Vp, por lo tanto si se desea que el resultado en la hoja de cálculo sea positivo, como en el ejemplo, se debe anteponer al nombre de la función el signo negativo (Ver barra de fórmulas)<sup>4</sup>.

**2.2.9 PAGO.PRINC.ENTRE**

Devuelve el monto de los abonos a capital efectuados entre dos fechas, para lo cual basta con dar los datos del crédito y el número ordinal de las cuotas a pagar donde empieza y termina el período

**PAGO.PRINC.ENTRE(tasa, nper, vp, per\_inicial, per\_final, tipo)**

Su operación es igual a la función anterior, pero se puede obtener una ventaja adicional: combinándola con el valor del crédito es posible saber cual es el saldo de la deuda en un momento determinado. Por ejemplo, si se desea saber cuál es el saldo de la deuda del caso anterior es posible encontrarlo de la siguiente manera:

<sup>4</sup> Debe aclararse que ni esta ni la función PAGO.PRINC.ENTRE debe emplearse cuando el pago de la cuota es anticipado, ya que presenta las mismas dificultades vistas en PAGOINT

	A	B	C
8			
9	SALDO INICIAL	\$ 50,000,000	
10	CUOTAS PAGADAS	\$ 36,176,901	=PAGO(3%,180,-50000000,0,0)*24
11	INTERESES PAGADOS	\$ 35,923,148	=-PAGO.INT.ENTRE(3%,180,50000000,1,24,0)
12	SALDO MES 24	\$ 49,746,246	=+B9-(B10-B11)
13			
14	SALDO CON FUNCION	\$ 49,746,246	
15			
16	PAGO.PRINC.ENTRE(tasa, nper, vp, per_inicial, per_final, tipo)		

Las dos funciones anteriores no utilizan el argumento VF por lo tanto no sirven para calcular por ejemplo cuánto se ha ganado en una inversión durante un determinado lapso de tiempo.

## 2.3 FUNCIONES DE EVALUACION DE PROYECTOS

La evaluación financiera de proyectos consiste en la aplicación de algunos indicadores de conveniencia económica al flujo de caja estimado de un negocio. En esta parte se presentarán las funciones financieras del Excel que pueden utilizarse como indicadores de conveniencia económica.

Es bien sabido que en un proyecto real el flujo de efectivo resultante no obedece a ninguna de las series conocidas (anualidades, gradientes, etc.), puesto que depende de gran cantidad de variables, por lo tanto no existe una fórmula para calcular el valor presente neto o la tasa de retorno, es necesario trabajar cada componente del flujo como un elemento independiente. Es aquí donde el Excel presenta un gran aporte para la evaluación financiera de proyectos.

Los argumentos que utilizan las funciones de evaluación de proyectos son los siguientes:

ARGUMENTO	SIGNIFICADO Y OBSERVACIONES
Tasa	Es la tasa de descuento que se utiliza para calcular el valor presente. Debe expresarse en el mismo período que se utiliza para la serie de datos.
Valor1, valor2,	Son los rangos que contienen los valores (ingresos y egresos) a los cuales se les va a calcular el valor presente; la función acepta hasta 29 rangos.
Valores	Es un rango que contiene los valores (flujo de caja) a los cuales se desea calcular la tasa interna de retorno. El argumento valores debe contener al menos un valor positivo y uno negativo para calcular la tasa interna de retorno. Estos flujos de caja no tienen por que ser constantes, como es el caso en una anualidad, sin embargo, los flujos de caja deben ocurrir en intervalos regulares.
Tasa_financiamiento	Es la tasa de interés que se paga sobre el dinero utilizado en los flujos de caja.
Tasa_reinversión	Es la tasa de interés a la que se reinvierten los flujos de caja del proyecto a medida que ingresan.
Fechas	Es un calendario de fechas de pago del flujo de caja. La primera fecha indica el principio del calendario, el resto de las fechas deben ser posteriores, pero pueden ocurrir a intervalos irregulares

### 2.3.1 VNA

Devuelve el valor presente de los flujos futuros de una inversión. Esta función **no** calcula el Valor Presente Neto (VPN)

### VNA(tasa, valor1, valor2, ...)

Los valores que se incluyen en el flujo de caja no tienen que ser constantes; esta es la principal diferencia frente a la función VA, pues aún se conserva la condición de que tanto la tasa de interés como la periodicidad sean constantes, es decir que todo el flujo de caja se descuenta a la misma tasa y los valores que se incluyen en él ocurren a intervalos iguales.

F7		=VNA(F4,D5:D16)						
	A	B	C	D	E	F	G	H
1								
2	MES	INGRESOS	EGRESOS	NETO				
3								
4	0		20,000,000	-20,000,000		3.0%	5.0%	<== Tasa
5	1	2,350,000	500,000	1,850,000				
6	2	2,350,000	500,000	1,850,000		-20,000,000	-20,000,000	<== Inversión inicial
7	3	2,350,000	1,250,000	1,100,000		23,025,506	19,908,812	<== Valor presente
8	4	2,350,000	650,000	1,700,000		3,025,506	-91,188	<== Valor presente neto
9	5	3,300,000	650,000	2,650,000				
10	6	3,300,000	4,500,000	-1,200,000				<== Valores
11	7	3,300,000	750,000	2,550,000				
12	8	4,350,000	750,000	3,600,000				
13	9	4,350,000	2,500,000	1,850,000				
14	10	3,850,000	850,000	3,000,000				
15	11	1,000,000	850,000	150,000				
16	12	10,000,000		10,000,000				

Dentro del rango del flujo de caja al que se le va a calcular el valor presente neto no debe incluirse el valor que se encuentra en el período cero, pues tal valor ya se encuentra en pesos de hoy. En el ejemplo anterior la inversión inicial que se encuentra en la celda D4 no se incluye en el argumento valores y posteriormente debe restarse del resultado que arroje la función para calcular el Valor Presente Neto (VPN)

El valor presente neto es un indicador sobre la conveniencia económica de una inversión que involucra la subjetividad del inversionista, ya que debe seleccionarse una tasa de interés para descontar el flujo de caja (3% en F4 o 5% en G4, como se aprecia en el ejemplo anterior). Como se ha calculado con dos tasas diferentes arroja dos resultados diferentes, para evaluar estos resultados debe tenerse en cuenta que la respuesta está expresada en pesos del período cero

De lo anterior se concluye que cuando se anuncie el VNA de un proyecto debe aclararse cuál fue la tasa de descuento que se utilizó para calcularlo, es decir cuál fue el valor que se empleó en el argumento Tasa.

### 2.3.2 TIR

Devuelve la rentabilidad (Tasa Interna de Rentabilidad) de una inversión, expresada en una tasa de interés equivalente al mismo período en que se presente la inversión.

#### TIR(valores, estimar)

Para el cálculo de la función TIR se incluye en el rango de valores todo el flujo de caja y es necesario que existan valores positivos y negativos. El argumento Estimar es opcional como se puede apreciar en la figura y en caso de omitirse, el Excel asume una tasa inicial del 10%.

En el caso del ejemplo anterior la TIR es igual a 4.94%, esta es la razón por la cual el VNA al 5% arrojó un valor negativo, indicando que la rentabilidad real del negocio era inferior a la solicitada por el inversionista.

Se deduce, entonces, que la TIR solo involucra las condiciones particulares de un proyecto y no se ve afectada por la subjetividad del inversionista. Sin embargo, presenta dificultades de orden matemático que llevan a desconfiar de los resultados que arroje. Para ilustrar el caso se presenta el siguiente ejemplo tomado del libro Evaluación Financiera de Proyectos de Inversión (pág. 114):

Como se aprecia, ante el mismo flujo de caja la función TIR arroja dos resultados diferentes, dependiendo del valor que se utilice en el argumento Estimar. Esta situación hace que sea de mucho cuidado la utilización de esta función pues puede llevar a conclusiones erróneas.

D18		=TIR(D4:D16)			
	A	B	C	D	E
1					
2	MES	INGRESOS	EGRESOS	NETO	
3					
4	0		20,000,000	-20,000,000	
5	1	2,350,000	500,000	1,850,000	
6	2	2,350,000	500,000	1,850,000	
7	3	2,350,000	1,250,000	1,100,000	
8	4	2,350,000	650,000	1,700,000	
9	5	3,300,000	650,000	2,650,000	
10	6	3,300,000	4,500,000	-1,200,000	
11	7	3,300,000	750,000	2,550,000	
12	8	4,350,000	750,000	3,600,000	
13	9	4,350,000	2,500,000	1,850,000	
14	10	3,850,000	850,000	3,000,000	
15	11	1,000,000	850,000	150,000	
16	12	10,000,000		10,000,000	
17					
18			TIR ==>	4.94%	

B8		=TIR(B3:B5,5%)			
	A	B	C	D	E
2		FLUJO			
3		(72,727)			
4		170,909			
5		(100,000)			
6					
7					
8		10.0%		=TIR(B3:B5,5%)	
9		TIR calculada con el argumento estimar igual a 5%			
10					
11					
12					
13		25.0%		=TIR(B3:B5,20%)	
14		TIR calculada con el argumento estimar igual a 20%			

### 2.3.3 TIRM

Devuelve la tasa interna de rentabilidad de un flujo de caja periódico teniendo en cuenta el costo en que se incurre para financiar los egresos del negocio y el beneficio obtenido por la reinversión de los ingresos que genera la inversión.


#### TIRM(valores, tasa\_financiamiento, tasa\_reinversión)

El principio en que se basa esta función es el del valor del dinero en el tiempo, que se desglosa en los dos puntos siguientes:


- a) Los egresos en que se incurre durante la vida de un proyecto tienen un costo de financiación por que el inversionista los debe reunir ahorrando, tomándolos en préstamo o retirándolos de otra inversión, por lo tanto este costo debe tenerse en cuenta en el cálculo de la rentabilidad de un proyecto. Para ello, los egresos del proyecto se descuentan al período cero (o se traen a valor presente) con una tasa de interés que refleje el costo de financiación de los recursos, representada por el argumento

tasa\_financiamiento. El valor presente de los egresos se calcula con una tasa de descuento prudente (igual a la tasa que rinden los depósitos

normales de la empresa) con el fin de saber en pesos de hoy cuánto se requeriría depositar en una cuenta para reunir los desembolsos que en cada período requiere el proyecto



- b) Los ingresos que recibe el inversionista durante la vida del proyecto pueden ser reinvertidos en otros negocios o en el mercado financiero. Para reconocer este proceso, los ingresos se llevan a valor futuro con una tasa de interés que refleje la tasa de oportunidad del inversionista, pues se supone que los dineros liberados por el proyecto se pueden reinvertir en los negocios a que normalmente tiene acceso el inversionista y no al rendimiento que genera este



proyecto en particular. La tasa de oportunidad esta representada por el argumento tasa\_reinversión

La TIR modificada (TIRM) esta mostrando cuál será la rentabilidad del proyecto para un inversionista en particular al relacionar el ingreso total que producirá (valor futuro de los ingresos) y con el monto que es necesario tener disponible hoy (valor presente de los egresos) para poder llevarlo a cabo.

El resultado que arroja la función TIRM es inferior al que se obtiene con la función TIR, pues esta última solo refleja las características del proyecto, mientras que la modificada incluye la subjetividad del inversionista al tener en cuenta el destino que dará a los recursos que va obteniendo del proyecto, tal como se aprecia a continuación:

F11		=TIRM(D4:D16,E20,E18)					
	A	B	C	D	E	F	G
1							
2	<b>MES</b>	<b>INGRESOS</b>	<b>EGRESOS</b>	<b>NETO</b>			
4	0		20,000,000	-20,000,000		<b>4.94%</b> <== TIR	
5	1	2,350,000	500,000	1,850,000		=TIR(D4:D16)	
6	2	2,350,000	500,000	1,850,000			
7	3	2,350,000	1,250,000	1,100,000			
8	4	2,350,000	650,000	1,700,000			
9	5	3,300,000	650,000	2,650,000			
10	6	3,300,000	4,500,000	-1,200,000			
11	7	3,300,000	750,000	2,550,000		<b>3.95%</b> <== TIRM	
12	8	4,350,000	750,000	3,600,000		=TIRM(D4:D16,E20,E18)	
13	9	4,350,000	2,500,000	1,850,000			
14	10	3,850,000	850,000	3,000,000			
15	11	1,000,000	850,000	150,000			
16	12	10,000,000		10,000,000			
17							

### 2.3.4 VNA.NO.PER

Devuelve el valor presente neto de un flujo de caja que no tiene periodicidad constante.

**VNA.NO.PER(tasa, valores, fechas)**

Con esta función se resuelve el limitante de la periodicidad constante que se ha mencionado para la función VNA. Es importante tener en cuenta dos aspectos en el cálculo del valor presente no periódico:

**Primero**, la tasa de interés que se utiliza como tasa de descuento debe estar expresada en términos efectivos anuales, para poder convertirla al número de días que haya entre un flujo y otro.

F14      =VNA.NO.PER(F12,E3:E9,A3:A9)					
A	B	C	D	E	F
1					
2	<b>FECHAS</b>	<b>DIAS</b>	<b>INGRESOS</b>	<b>EGRESOS</b>	<b>NETO</b>
3	30-jun-08			10,000,000	-10,000,000
4	30-sep-08	92	2,350,000	500,000	1,850,000
5	15-oct-08	107	2,075,000	500,000	1,575,000
6	30-ene-09	214	1,800,000	1,250,000	550,000
7	20-jun-09	355	1,525,000	650,000	875,000
8	15-oct-09	472	3,300,000	650,000	2,650,000
9	30-dic-09	548	12,000,000	4,500,000	7,500,000
10					<b>766,775</b>
11					
12	TASA DE DESCUENTO EFECTIVA ANUAL ==>				<b>36.0%</b>
13					
14	VALOR PRESENTE NETO NO PERIODICO ==>				<b>766,775</b>
15	=VNA.NO.PER(E13,E4:E10,A4:A10)				
16					
17	TIR NO PERIODICA ==>				<b>46.2%</b>
18	=TIR.NO.PER(E4:E10,A4:A10)				

**Segundo**, en el argumento valores se debe incluir todo el flujo de caja, no solo los valores que se encuentran en períodos futuros.

### 2.3.5 TIR.NO.PER

Devuelve la tasa interna de retorno para un flujo de caja que no tiene periodicidad constante.

#### TIR.NO.PER(valores, fechas, Estimar)

Para esta función se deben tener en cuenta las observaciones efectuadas en la función del valor presente no periódico, especialmente no olvidar que el resultado se expresa en términos efectivos anuales y que siempre utiliza para los cálculos el año de 365 días.

## 2.4 FUNCIONES DE INVERSIONES FINANCIERAS

Estas funciones sirven para efectuar cálculos en los cuales están involucrados papeles con características variadas en cuanto a la forma de pago de los rendimientos: con descuento, periódicamente o al final de la inversión. Las funciones clasificadas dentro de este grupo presentan diferencias frente a las costumbres del mercado financiero colombiano; algunas de estas diferencias son irrelevantes y se pueden conciliar fácilmente a través de la hoja de cálculo, pero hay otras diferencias que son considerables y por lo tanto hacen inoperantes algunas funciones en el mercado financiero colombiano. Por tales razones debe tenerse especial cuidado con la interpretación de los resultados que arrojen estas funciones.

Las funciones de inversiones financieras utilizan los siguientes argumentos:

ARGUMENTO	SIGNIFICADO Y OBSERVACIONES												
Emisión	Es la fecha de emisión de la inversión en el mercado primario.												
Liq	Es la fecha de compra de la inversión en el mercado secundario.												
Vencto	Es la fecha de vencimiento de la inversión.												
Inversión	Es la cantidad de dinero que se ha invertido en el instrumento financiero. En este caso no es necesario que se cambien los signos entre ingresos y egresos, como si se debe hacer en los problemas de series uniformes.												
Valor_de_rescate	Es el valor nominal de una inversión comprada con descuento, es decir el valor que se recibirá en la fecha de vencimiento. Normalmente se utiliza \$ 100.												
Base	<p>Determina el tipo de base en que deben ser contados los días. Excel utiliza cinco tipos de bases con los siguientes códigos que pueden utilizarse en el argumento:</p> <table style="margin-left: auto; margin-right: auto;"> <thead> <tr> <th><u>Base (Código)</u></th> <th><u>Número de días que utiliza</u></th> </tr> </thead> <tbody> <tr> <td>0 u omitida</td> <td>30/360</td> </tr> <tr> <td>1</td> <td>Real/real</td> </tr> <tr> <td>2</td> <td>Real/360</td> </tr> <tr> <td>3</td> <td>Real/365</td> </tr> <tr> <td>4</td> <td>Europea 30/360</td> </tr> </tbody> </table>	<u>Base (Código)</u>	<u>Número de días que utiliza</u>	0 u omitida	30/360	1	Real/real	2	Real/360	3	Real/365	4	Europea 30/360
<u>Base (Código)</u>	<u>Número de días que utiliza</u>												
0 u omitida	30/360												
1	Real/real												
2	Real/360												
3	Real/365												
4	Europea 30/360												
Tasa	Es la tasa de interés nominal anual.												
V_nominal	Es el precio inscrito en el documento, representativo del valor a pagar a su vencimiento.												
1er_interés	Es la fecha en que se pagará el próximo interés, después de la fecha de compra en el mercado secundario; es decir cuando el nuevo propietario recibirá los primeros intereses en un título con rendimientos periódicos.												
FREC	Es el número de períodos de capitalización que hay en un año. Es												

ARGUMENTO	SIGNIFICADO Y OBSERVACIONES
	posible utilizar frecuencia anual (1), semestral (2) y trimestral (4).
Descuento	Es la tasa de descuento nominal anual a la que se hace el negocio. Hay que tener en cuenta que <u>no es</u> el porcentaje que el vendedor descontó del valor futuro del título.
Rendto	Rendimiento que espera obtener la persona que compra una inversión en el mercado secundario.
Precio	Es el precio de compra de una inversión con descuento.
Capital	Es el valor inicial de una inversión.
Plan_serie_de_tasas	Es un rango con las tasas de interés que se aplican a una inversión que capitaliza intereses variables.

La fecha de emisión es la fecha en que empieza el bono.

La fecha de liquidación (liq) es la fecha en que se compra un bono.

La fecha de vencimiento es la fecha en que termina el bono.

Por ejemplo, un bono emitido a cinco años el 1 de enero de 2009 lo adquiere un inversionista nueve meses más tarde.

La fecha de emisión será el 1 de enero de 2009

La fecha de liquidación será el 1 de octubre de 2009

La fecha de vencimiento será el 31 de diciembre de 2013

### 2.4.1 TASA.INT

Devuelve la tasa de interés nominal anual liquidada diariamente para una inversión con pago único al final del período.

#### TASA.INT(liq, vencio, inversión, valor\_de\_rescate, base)

Esta función trabaja con interés simple durante todo el plazo de la inversión, es decir sin capitalización de intereses durante el tiempo que dure..

	A	B	C	D
1	FECHA DE LIQUIDACION	15-oct-09		
2	FECHA DE VENCIMIENTO	30-mar-10		
3	INVERSION	890,000		
4	VALOR DE RESCATE	1,000,000		
5				
6	RENTABILIDAD	26.9663%		
7				
8	SOLUCION MANUAL			
9				
10	RENTABILIDAD DEL PERIODO	12.36%	=B4/B3-1	
11	NUMERO DE DIAS DEL PERIODO	165	=DIAS360(B1,B2)	
12	RENTABILIDAD ANUAL	26.9663%	=B10/B11""360"	

### 2.4.2 INT.ACUM.V

Devuelve el monto en pesos del interés acumulado en una inversión con pago único al final del período, sin capitalización de intereses durante el tiempo que dure la inversión.

### INT.ACUM.V(emisión, vencio, tasa, v\_nominal, base)

Se emplea en esta función el concepto de interés simple ya que se considera que solo hay un período de pago durante la vida de la inversión; esto quiere decir que en nuestro medio no puede utilizarse por ejemplo cuando se presenta una inversión a seis meses y la tasa de interés se expresa en pagos mensuales, como se aprecia a continuación:

B9		=INT.ACUM.V(B2,B3,B5,B6,B7)	
A	B	C	
1			
2	EMISION	01-ene-09	
3	VENCIMIENTO	30-jun-09	
4	MESES (n)	6	
5	TASA	24%	NAMV (PAGO AL FINAL CON CAPITALIZACION MENSUAL)
6	VALOR NOMINAL	1,000,000	
7	BASE	0	
8			
9		119,333	<=<INTERES ACUMULADO SIMPLE UTILIZANDO LA FUNCION
10			
11			
12			
13	CALCULO MANUAL		
14	INTERES	119,333	=B6*B5/12*B4
15			
16			
17		125,419	<=<INTERES ACUMULADO COMPUESTO UTILIZANDO LA FORMULA I = F - P I = P(1+ln)^n - P
18			
19			
20			

#### 2.4.3 INT.ACUM

Devuelve la suma de los intereses pagados por una inversión desde su emisión hasta la fecha de venta (liq).

### INT.ACUM(emisión, 1er\_interés, liq, tasa, v\_nominal, frec, base)

El argumento 1er\_interés es la fecha en que se realizará el próximo pago de intereses después de la fecha de venta de la inversión y no el primer pago de intereses después de la emisión.

Esta función no es de utilidad, ya que sumar los intereses liquidados y pagados por una inversión en el pasado no aporta en el análisis de inversiones.

#### 2.4.4 PRECIO.DESCUENTO

Devuelve el valor que debe pagarse por una inversión que se negocia a una determinada tasa de descuento.

B8		=INT.ACUM(B1,B2,B3,B4,B5,B6,0)	
A	B	C	
1	EMISION	15-mar-02	
2	PROXIMO PAGO	15-sep-09	
3	FECHA LIQUIDACION	20-ago-09	
4	TASA NOMINAL ANUAL	18%	
5	VALOR NOMINAL	1,000,000	
6	FRECUENCIA	4 TRIMESTRAL	
7			
8	INTERES ACUMULADO	1,337,500	
9			
10	TASA PERIODICA	4.5%	=B4/B6
	NUMERO DE TRIMESTRES TRANSCURRIDOS DESDE LA EMISION	29.7	=DIAS360(B1,B3)/90
11	INTERES ACUMULADO HASTA LA FECHA DE LIQUIDACION	1,337,500	=B10*B11*B5
12			

### PRECIO.DESCUENTO(liq, vencio, descuento, valor\_de\_rescate, base)

En caso de no conocerse el valor de rescate de la inversión puede utilizarse el número uno (1) y la respuesta estaría dada en porcentaje

Esta función tampoco es de utilidad en nuestro medio, ya que utiliza interés simple. Como se aprecia es el caso típico de un bono cero cupón que se vende con descuento, como se observa en el ejemplo.

B7		=PRECIO.DESCUENTO(B1,B2,B3,B4,0)	
	A	B	C
1	FECHA LIQUIDACION	20-ago-09	
2	FECHA VENCIMIENTO	30-dic-12	
3	DESCUENTO	12.0%	
4	VALOR DE RESCATE	1,000,000	
5			
6			
7	PRECIO DESCUENTO	596,667	
8			
9	CALCULO MANUAL		
10	PLAZO EN DIAS	1,210	=DIAS360(B1,B2)
11	TASA DEL PLAZO	40.33%	=B3*B10/360
12	INTERESES PENDIENTES	403,333	=B4*B11
13	PRECIO DE LA INVERSION	596,667	=B4-B12
14			
15			
16	CALCULO CON INTERES COMPUESTO		
17	PLAZO EN DIAS	1,210	=B10
18	PRECIO DE LA INVERSION	683,239	=B4/(1+B3)^(B17/360)

La formula empleada por esta función es la siguiente:

$$\text{PRECIO} = \text{VALOR DE RESCATE} * \left( 1 - i \frac{\text{DIAS}}{360} \right)$$

Mientras la formula utilizada en la valoración de bonos es:

$$\text{PRECIO} = \frac{\text{VALOR DE RESCATE}}{(1 + i)^{\text{DIAS}/360}}$$

## 2.4.5 RENDTO.DESC

Devuelve el rendimiento nominal anual de una inversión negociada con descuento. Solo sirve cuando el descuento es la única rentabilidad que ofrece la inversión.

**RENDTO.DESC(liq, vencto, precio, valor\_de\_rescate, base)**

Esta función utiliza la siguiente formula de interés simple:

$$\text{TASA} = \frac{\text{PRECIOVENTA} - \text{PRECIOCOMPRA}}{\text{PRECIOCOMPRA}} * \frac{360}{\text{DIAS}}$$

B6		=RENDTO.DESC(B1,B2,B3,B4,0)	
	A	B	C
1	FECHA LIQUIDACION	20-ago-09	
2	VENCIMIENTO	30-dic-12	
3	PRECIO	596,667	
4	VALOR DE RESCATE	1,000,000	
5			
6	RENDIMIENTO	20.1117%	
7			
8	CALCULO MANUAL		
9	PLAZO EN AÑOS	3.36	=DIAS360(B1,B2)/360
10	INVERSION	596,667	=B3
11	RESCATE	1,000,000	=B4
12	UTILIDAD	403,333	=B11-B10
13	RENDIMIENTO TOTAL	67.60%	=B12/B10
14	RENDIMIENTO ANUAL	20.1117%	=B13/B9
15			
16			
17	CALCULO CON INTERES COMPUESTO		
18	PLAZO EN AÑOS	3.36	=B9
19	INVERSION	596,667	=B3
20	RESCATE	1,000,000	=B4
21	RENDIMIENTO	16.61%	=(B20/B19)^(1/B18)-1

Con interés compuesto se utiliza la siguiente formula:

$$TASA = \left( \frac{PRECIOVENTA}{PRECIOCOMPRA} \right)^{360/DIAS}$$

Esta es la formula para calcular la rentabilidad al vencimiento de una inversión con descuento sin pago de intereses (cupón cero)

#### 2.4.6 CANTIDAD.RECIBIDA

Calcula el monto de dinero que recibirá un inversionista al vencimiento de una inversión comprada con descuento, si la conserva hasta el final.

#### **CANTIDAD.RECIBIDA(liq, vencto, inversión, descuento, base)**

Para calcular este valor, se supone que la inversión es solo con descuento y no paga otros rendimientos durante el tiempo que se conserve. No sobra repetir que el argumento Descuento esta expresado en términos anuales y no se trata del descuento que le hizo el vendedor al valor nominal del título para poder venderlo. De ahí que la fórmula empleada para calcular la Cantidad.Recibida sea la siguiente:

$$PV = \frac{PC}{1 - \left[ \frac{\text{descuento}}{360} * \text{dias} \right]}$$

Donde:

PV Precio de venta o cantidad recibida

PC Precio de compra o inversión

descuento Tasa de descuento nominal anual

dias Número de dias que hay entre la fecha de compra y la fecha de vencimiento de la inversión

Como se aprecia, en el contexto colombiano esta función no tiene mayor aplicación, ya que en las inversiones financieras no se utiliza la tasa de descuento anual equivalente; en el mercado colombiano se utiliza o bien el descuento efectivo por todo el período de

la inversión o más comúnmente el precio de venta del papel expresado en términos porcentuales (1-descuento) y por lo tanto la fórmula empleada es más sencilla:

$$PV = \frac{PC}{1 - \text{descuento}} = \frac{PC}{\text{precio}}$$

A continuación se presenta un ejemplo para evitar errores: Una Aceptación Bancaria que se adquiere por \$ 12.530.000 (el 92% de su valor nominal) y le faltan 90 días para su vencimiento:

	B	C	D	E	F	G
1						
2				CANTIDAD RECIBIDA		
3	FECHA COMPRA	12-Jun-98		UTILIZANDO LA FUNCION ==>	\$ 13,619,565	
4	FECHA VENCIMIENTO	12-Sep-98				
5	INVERSION	\$ 12,530,000				
6	DESCUENTO	32%	<== EL DESCUENTO DEBE SE ANUALIZADO			
7						
8	PLAZO INVERSION	90		CANTIDAD RECIBIDA UTILIZANDO		
9				LA FORMULA DE LA FUNCION ==>	\$ 13,619,565	C5/(1-C6/360*C8)
10						
11						
12	PRECIO	92%		CANTIDAD RECIBIDA UTILIZANDO		
13	DESCUENTO	8%		FORMULA SENCILLA ==>	\$ 13,619,565	C5/C12
14						

Con las condiciones anteriores es que se anuncia el papel en Colombia, es decir con un descuento del 8% sobre su valor nominal; pero en las funciones del Excel, este descuento se debe anualizar, como se aprecia en el ejemplo, y en la función se debe emplear el 32%, tal como figura en la celda C6.

Otro problema que se presenta es que se deben conocer las fechas exactas de compra y vencimiento de los papeles y no basta con conocer el plazo de la inversión.

### 2.4.7 TASA.DESC

Devuelve la tasa de descuento en su equivalente nominal anual de una inversión comprada con descuento.

**TASA.DESC(liq, vencio, precio, valor\_de\_rescate, base)**

Esta función presenta las mismas características de la anterior y por lo tanto no tiene mayor aplicación en el mercado financiero colombiano, donde interesa la rentabilidad efectiva anual de la inversión y no la tasa nominal anual del descuento.

### 2.4.8 CUPON.NUM

Devuelve el número de pagos de capital entre la fecha de compra (liq) y la fecha de vencimiento de una inversión, redondeados al número entero del pago más cercano.

#### CUPON.NUM(liq, vencto, frec, base)

En terminología de los bonos, calcula el número de cupones que recibirá quien compre el bono en la fecha de liquidación y lo conserve hasta el vencimiento.

Es una función útil para trabajar problemas de bonos

	A	B	C
1	FECHA LIQUIDACION	01/sep/2009	
2	FECHA VENCIMIENTO	30/dic/2012	
3	FRECUENCIA DEL CUPON	2 semestral	
4			
5	Fecha de pago de los cupones		
6	30/dic/2009		1
7	30/jun/2010		2
8	30/dic/2010		3
9	30/jun/2011		4
10	30/dic/2011		5
11	30/jun/2012		6
12	30/dic/2012		7
13			
14	CUPON NUMERO		7

### 2.4.9 CUPON.DIAS

Devuelve el número de días que transcurren entre dos pagos de interés o de capital en una inversión con pagos periódicos.

#### CUPON.DIAS(liq, vencto, frec, base)

Es una función inútil, pues devuelve un dato que previamente se conoce.

### 2.4.10 CUPON.DIAS.L1

Calcula el número de días que han transcurrido entre el último pago de intereses y la fecha de venta (liq) de una inversión.

#### CUPON.DIAS.L1(liq, vencto, frec, base)

Función de utilidad para calcular la valorización acumulada del cupón desde la fecha del último pago.

B5 =CUPON.DIAS.L1(B1,B2,B3,0)			
	A	B	C
1	FECHA LIQUIDACION	01/sep/2009	
2	FECHA VENCIMIENTO	30/dic/2012	
3	FRECUENCIA DEL CUPON	2 semestral	
4			
5	DIAS TRANSCURRIDOS DESDE EL ÚLTIMO PAGO DE CUPON	61	

### 2.4.11 CUPON.DIAS.L2

Calcula el número de días que faltan para el próximo pago de intereses, contados desde la fecha de venta (liq).

#### CUPON.DIAS.L2(liq, vencto, frec, base)

Función muy útil para calcular la rentabilidad que podrá esperar el inversionista.

B5 =CUPON.DIAS.L2(B1,B2,B3,0)			
	A	B	C
1	FECHA LIQUIDACION	01/sep/2009	
2	FECHA VENCIMIENTO	30/dic/2012	
3	FRECUENCIA DEL CUPON	2 semestral	
4			
5	DIAS QUE FALTAN PARA EL PROXIMO PAGO DE CUPON	119	

### 2.4.12 CUPON.FECHA.L1

Calcula la fecha en la cual se pagaron los últimos intereses, previos a la fecha de la negociación del papel. Esta fecha se expresa en el número de serie, por lo tanto hay que darle formato de fecha para que sea entendible.

#### CUPON.FECHA.L1(liq, vencto, frec, base)

Función de utilidad para calcular la valorización acumulada del nuevo cupón desde la fecha del último pago.

B5 =CUPON.FECHA.L1(B1,B2,B3,0)			
	A	B	C
1	FECHA LIQUIDACION	01/sep/2009	
2	FECHA VENCIMIENTO	30/dic/2012	
3	FRECUENCIA DEL CUPON	2 semestral	
4			
5	FECHA DE PAGO DEL CUPON ANTERIOR	30/jun/2009	

### 2.4.13 CUPON.FECHA.L2

Calcula la fecha en que se pagarán los próximos de intereses después de la fecha de venta (liq) de una inversión.

#### CUPON.FECHA.L2(liq, vencto, frec, base)

A continuación se presenta un ejemplo con las anteriores funciones, las cuales, a pesar de ser tan sencillas, son de gran utilidad en el análisis de inversiones que tienen pagos periódicos de capital o de intereses:

B11		=CUPON.FECHA.L2(B1,B2,B3,B4)	
	A	B	C
1	FECHA DE VENTA	23-Abr-98	
2	FECHA DE VENCIMIENTO	15-Sep-99	
3	FRECUENCIA	4	
4	BASE	0	
5			
6	NUMERO DE PAGOS DE INTERES QUE FALTAN	6	=CUPON.NUM(B2,B3,B4,B5)
7	DIAS ENTRE DOS PAGOS DE INTERES	90	=CUPON.DIAS(B2,B3,B4,B5)
8	DIAS DE INTERES CAUSADOS EN EL PRESENTE PERIODO	38	=CUPON.DIAS.L1(B2,B3,B4,B5)
9	DIAS QUE FALTAN PARA EL PAGO EN EL PRESENTE PERIODO	52	=CUPON.DIAS.L2(B2,B3,B4,B5)
10	FECHA DEL ANTERIOR PAGO DE INTERESES	15-Mar-98	=CUPON.FECHA.L1(B2,B3,B4,B5)
11	FECHA DEL PROXIMO PAGO DE INTERESES	15-Jun-98	=CUPON.FECHA.L2(B2,B3,B4,B5)
12			

#### 2.4.14 VF.PLAN

Devuelve el valor futuro de un capital inicial después de aplicar una serie de tasas de interés compuesto.

#### VF.PLAN(capital, plan\_serie\_de\_tasas)

Es una función de gran utilidad para simular un valor futuro que capitaliza con tasas variables, como se aprecia en el siguiente ejemplo:

D5		=VF.PLAN(B3,B6:B9)	
	A	B	C
1			
2			
3	CAPITAL	1,250,000	
4			
5	MES	TASA	1,391,835
6	1	3.0%	
7	2	2.0%	=B3*(1+B5)*(1+B6)*(1+B7)*(1+B8)
8	3	3.5%	
9	4	2.4%	
10			


#### 2.4.15 PRECIO.VENCIMIENTO

Devuelve el precio máximo, por cada \$100 de valor nominal, al que debe comprarse una inversión, para que en el tiempo restante hasta el vencimiento arroje un rendimiento deseado por el comprador.

#### PRECIO.VENCIMIENTO(liq, vencio, emisión, tasa, rendto, base)

Esta función se aplica solo a inversiones que pagan todo el rendimiento al final del plazo cualquiera que sea su duración y trabaja con interés simple.

No es aplicable directamente en el medio colombiano debido a que la respuesta está expresada en términos del valor nominal de la inversión y aunque tiene en cuenta los intereses causados hasta la fecha de la negociación, no los expresa en la respuesta. Pero es relativamente sencillo efectuar el ajuste utilizando el siguiente gráfico:



Del gráfico anterior se deduce que si la función expresa su respuesta en términos del valor nominal, para quedar expresada en términos del valor actualizado solo basta con sumarle los intereses causados desde la fecha de emisión hasta la fecha de negociación, los cuales se obtienen con la función Int.Acum.V (ver pág. 30). Una vez realizado este pequeño ajuste el precio de venta será igual al valor absoluto que se debe pagar por un papel que ha causado intereses desde su emisión.

A continuación se ilustra esta función con el siguiente ejemplo:

**Por cuanto se debe comprar un CDT a 90 días, emitido el 2 de mayo que rinde el 12%, si el 4 de junio un comprador en el mercado secundario desea tener una rentabilidad del 15% por el plazo restante**

Utilizando la combinación de funciones mencionada, en la hoja de cálculo Excel la solución sería la siguiente:

	A	B	C
1	FECHA LIQUIDACION	04/jun/2009	
2	FECHA VENCIMIENTO	02/ago/2009	
3	FECHA DE EMISION	02/may/2009	
4	TASA	12%	
5	RENDIMIENTO	15%	
6	VALOR NOMINAL	100	
7			
8	<b>PRECIO</b>	<b>99.50</b>	
9			
10	DIAS DE LA INVERSION	90	
11	VALOR AL VENCIMIENTO	103.0	
12			
13	VALOR NOMINAL	100.00	
14	INTERESES CAUSADOS	1.07	
15	<b>VALOR ACTUALIZADO</b>	<b>101.07</b>	
16			
17	DIAS QUE FALTAN AL VENCIMIENTO	58	
18	PRECIO + INTERESES	100.57	
19			
20	RENTABILIDAD EN EL PLAZO RESTANTE	2.42%	
21	RENTABILIDAD ANUAL	15.0%	

La función directamente arroja un resultado de 99.5% (celda B8) que en el mercado colombiano no tiene ningún significado, pero al sumarle los

intereses causados desde la emisión hasta la fecha (celda B14) se tiene la respuesta correcta, como se aprecia en la celda B21. Reinterpretando el resultado se dirá que por cada \$100 de valor nominal de la inversión se deben pagar máximo \$99.50 más los intereses causados, para obtener la rentabilidad deseada de 15%.

#### 2.4.16 RENDTO.VENCTO

Devuelve el rendimiento nominal anual por el tiempo que resta del plazo, de una inversión comprada en el mercado secundario.

#### RENDTO.VENCTO(liq, vencto, emisión, tasa, precio, base)

A continuación se ilustra esta función con el mismo ejemplo que utilizado en la función anterior, por lo tanto debe arrojar los mismos resultados:

B8		=RENDTO.VENCTO(B1,B2,B3,B4,B5,0)		
	A	B	C	
1	FECHA LIQUIDACION	04/jun/2009		
2	FECHA VENCIMIENTO	02/ago/2009		
3	FECHA DE EMISION	02/may/2009		
4	TASA	12%		
5	PRECIO	99.5		
6	VALOR NOMINAL	100		
7				
8	RENDIMIENTO			15.00%

Al igual que la función anterior, ésta solo se aplica a inversiones que pagan todo el rendimiento al final del plazo. Tampoco puede utilizarse directamente en el medio colombiano.

#### 2.4.17 PRECIO

Devuelve el precio máximo, con respecto al valor nominal, al que debe comprarse una inversión que paga intereses periódicos, para que en el tiempo restante, hasta el vencimiento, arroje un rendimiento deseado por el comprador.

#### PRECIO(liq, vencto, tasa, rendto, valor\_de\_rescate, frec, base)

Adecuar esta función a la costumbre del mercado financiero colombiano, de expresar los precios en términos del valor actualizado de la inversión, es más exigente debido a que es necesario conocer dos datos que no están involucrados en los argumentos originales como son la fecha de emisión del papel y la fecha del próximo pago de intereses. Conociendo estos datos es posible calcular los intereses que se han causado desde el inicio del período actual de pago de intereses hasta la fecha (celda D4) y

sumarlos al precio calculado por la función (celda A10) para obtener el precio en términos del valor actualizado (celda D7).

	A	B	C	D	E
1	COMPRA	18-Sep-98			
2	VENCE	02-May-99			
3	TASA	27.00%			
4	RENTA	33.39%		10.2000	INTERES CAUSADO HASTA LA FECHA
5	RESCATE	100.0	<== VALOR NOMINAL		=INT.ACUM(A7,A8,B1,B3,B5,B6,0)
6	FRECUENCIA	4			
7	02-May-98	EMISION		106.6687	PRECIO DE COMPRA
8	02-Nov-98	PROXIMO PAGO DE INTERESES			=A10+D4
9					
10	96.4687	PRECIO SOBRE VALOR NOMINAL		33.39%	RENTABILIDAD
11		=PRECIO(B1,B2,B3,B4,B5,B6,0)			=RENDTO(B1,B2,B3,D7-D4,B5,B6,0)
12					

## 2.4.18 RENDTO

Calcula el rendimiento nominal anual, desde la fecha de compra en el mercado secundario hasta la fecha de vencimiento, de un valor bursátil que paga intereses periódicos.

**RENDTO(liq, vencio, tasa, precio, valor\_de\_rescate, frec, base)**

En el ejemplo anterior, como se observa en la celda A19, si en el argumento precio se desea utilizar el precio sobre el valor actualizado, deben restársele los intereses del período actual de pago acumulados hasta la fecha de compra en el mercado secundario.

## 2.5 FUNCIONES PARA MEDIR EL RIESGO DE LAS INVERSIONES

Las funciones para medir el riesgo de las inversiones utilizan los siguientes argumentos:

ARGUMENTO	SIGNIFICADO Y OBSERVACIONES												
Liq	Es la fecha de corte de la evaluación del riesgo												
Vencto	Es la fecha de terminación del bono												
Cupon	Es la tasa cupón del bono vigente al momento de evaluar el riesgo . Se expresa en nominal anual y la periodicidad será igual a la frecuencia de pago de los cupones												
Rendto	Es la tasa de rentabilidad existente en el mercado para papeles de riesgo similar al que se está evaluando												
Frec	Periodicidad de pago de los cupones. Este argumento solo acepta los siguientes valores: 1 (anual), 2 (semestral) y 4 (trimestral).												
Base	Determina el tipo de base en que deben ser contados los días. Excel utiliza cinco tipos de bases con los siguientes códigos que pueden utilizarse en el argumento: <table border="1" data-bbox="487 1008 1218 1260"> <thead> <tr> <th><u>Base (Código)</u></th> <th><u>Número de días que utiliza</u></th> </tr> </thead> <tbody> <tr> <td>0 u omitida</td> <td>30/360</td> </tr> <tr> <td>1</td> <td>Real/real</td> </tr> <tr> <td>2</td> <td>Real/360</td> </tr> <tr> <td>3</td> <td>Real/365</td> </tr> <tr> <td>4</td> <td>Europea 30/360</td> </tr> </tbody> </table>	<u>Base (Código)</u>	<u>Número de días que utiliza</u>	0 u omitida	30/360	1	Real/real	2	Real/360	3	Real/365	4	Europea 30/360
<u>Base (Código)</u>	<u>Número de días que utiliza</u>												
0 u omitida	30/360												
1	Real/real												
2	Real/360												
3	Real/365												
4	Europea 30/360												

### 2.5.1 DURACION

Devuelve el tiempo promedio, expresado en años, en el cual se espera recibir los flujos de efectivo de un instrumento financiero o lo que es lo mismo el tiempo promedio que se tarda en recuperar una inversión

**DURACION(liq, vencto, cupón, rendto, frec, base)**

La duración se define como el promedio ponderado de los períodos que faltan para el vencimiento de una inversión y la ponderación que se utiliza es el valor presente de los flujos futuros como porcentaje del valor de mercado del papel; la formula es la siguiente:

Donde:

$$D = \frac{\sum_{i=1}^n \frac{F_i}{(1+r)^{T_i}} * T_i}{P}$$

**D** Duración del bono

**F<sub>i</sub>** Flujo del bono en el periodo *i*

**r** Tasa diaria de rentabilidad del mercado

**P** Valor de mercado del bono (igual a la suma de los valores presentes de los flujos futuros del bono)

**T** Dias que faltan para recibir el flujo

En Colombia es obligación para los establecimientos financieros utilizar el concepto de duración para medir el riesgo de tasa de interés. Este riesgo consiste en los cambios que pueda tener el patrimonio de una entidad ante variaciones de la tasa de interés, pues bien se sabe que el plazo de las captaciones es considerablemente inferior al plazo de las colocaciones, por lo tanto un aumento en la tasa de interés puede encarecer los pasivos, antes que el ingreso reaccione por la reprecación de los activos, produciendo pérdidas.

A continuación se presenta un ejemplo para un título que vence el 31 de diciembre de 2011 y paga intereses del 6% NASV; la fecha de corte 30 de marzo de 2009 y la tasa de interés de mercado para ese mismo tipo de instrumento es del 8% NASV:

D9		=DURACION(D1,D2,D3,D4,D6,D7)					
	A	B	C	D	E	F	G
1	FECHA DE LIQUIDACION			30/mar/2009			
2	FECHA DE VENCIMIENTO			31/dic/2011			
3	TASA CUPON			6.0% NASV			
4	RENDIMIENTO DEL MERCADO			8.0% NASV			
5	RENDIMIENTO DEL MERCADO (r)			0.0218% diario			
6	FRECUENCIA			2			
7	BASE			0			
8							
9			DURACION	2.5331			
10							
11	<b>CALCULO MANUAL</b>						
12	Periodo (i)	Fecha	Dias (T <sub>i</sub> )	Flujo (F <sub>i</sub> )	Factor (1+r) <sup>T<sub>i</sub></sup>	VP flujos F <sub>i</sub> / Factor	VP flujos * T <sub>i</sub>
13	0	30-mar-09		(96.6)			
14	1	30-jun-09	90	3.0	1.01980	2.941742	264.76
15	2	31-dic-09	270	3.0	1.06060	2.828598	763.72
16	3	30-jun-10	450	3.0	1.10302	2.719806	1,223.91
17	4	31-dic-10	630	3.0	1.14714	2.615198	1,647.57
18	5	30-jun-11	810	3.0	1.19303	2.514613	2,036.84
19	6	31-dic-11	990	103.0	1.24075	83.014481	82,184.34
20							
21					P	96.634439	
22					SUMATORIA de VP * T <sub>i</sub>		88,121.14
23					DURACION - dias (SUMATORIA / P)		911.90
24					DURACION -años		2.5331

La Duración se utiliza para calcular cuánto cambia el precio de un bono cuando cambia la rentabilidad del mercado. Bien se sabe que cuando la rentabilidad del mercado aumenta, el precio de los papeles de renta fija disminuye (deben reducir el precio para poder ofrecer la misma rentabilidad que se está obteniendo en el mercado)

Para medir este impacto la formula que se utiliza es la siguiente:

$$\frac{\Delta P}{P} = \frac{-D}{1+c} \Delta r$$

$$P + \Delta P = P \left( 1 - \frac{D}{1+c} \Delta r \right)$$

- D** Duración del bono
- $\Delta r$**  Variación rentabilidad exigida al bono
- c** Tasa original del bono
- P** Precio de compra del bono
- $\Delta P$**  Variación precio del bono

Para el bono que se está analizando los resultados son los siguientes:

Estos resultados se interpretan de la siguiente manera: por cada 100 puntos básicos que aumente la tasa de mercado, el precio del bono disminuye 2.39%, indicando el riesgo que corre la empresa poseedora del bono ante un aumento de la tasa de mercado.

	A	B	C	D
				D72 =D68*(1-D71/(1+D69)*D70)
68	<b>Precio actual</b>			<b>96.63</b>
69	<b>Tasa nominal del bono</b>			<b>6.0%</b>
70	<b>Cambio en la tasa de mercado</b>			<b>1.00%</b>
71	<b>Duración</b>			<b>2.5</b>
72	<b>Nuevo precio</b>			<b>94.33</b>
73	<b>Variación porcentual del precio</b>			<b>(2.39%)</b>



### 3 BIBLIOGRAFÍA

Corredores Asociados. **Manual para el cálculo de rentabilidades**. Bogotá – 2005

García Jaime. **Matemáticas financieras**. Bogotá, Pearson – 2008

Gutiérrez, Jairo. **Matemáticas financieras con Excel**. En [www.jg-plan.net](http://www.jg-plan.net)

Gutiérrez, Jairo. **Modelos financiero con Excel**. Bogotá, ECOE – 2008

Hayad, S y San Millán, A. **Finanzas con Excel**. Madrid, McGraw-Hill – 2001

Hewlett Packard. **Manual del propietario** - Calculadora HP-17bII+

Martínez-Abascal, Eduardo. **Gestión de carteras de renta fija**. Madrid, McGraw-Hill – 2002

Meza, Jhonny. **Matemáticas financieras aplicadas**. Bogotá, ECOE – 2008

Pinilla, Roberto. **Operaciones financieras en el mercado bursátil**. Bogotá, Uniexternado – 2003

Superintendencia Financiera. **Resolución No. 0001** de enero 2 de 1996

#### 4 ANEXO N. 1

A continuación se presentan las funciones financieras del Excel, trabajadas en este fascículo, en orden alfabético:

CUPON.DIAS(liq, vencido, frec, base)	pág. 35
CUPON.DIAS.L1(liq, vencido, frec, base)	pág. 35
CUPON.DIAS.L2(liq, vencido, frec, base)	pág. 36
CUPON.FECHA.L1(liq, vencido, frec, base)	pág. 36
CUPON.FECHA.L2(liq, vencido, frec, base)	pág. 36
CUPON.NUM(liq, vencido, frec, base)	pág. 35
DURACION(liq, vencido, cupón, rendto, frec, base)	pág. 41
INT.ACUM(emisión, 1er_interés, liq, tasa, v_nominal, frec, base)	pág. 31
INT.ACUM.V(emisión, vencido, tasa, v_nominal, base)	pág. 30
INT.EFECTIVO(int_nominal, núm_per_año)	pág. 8
NPER(tasa, pago, va, vf, tipo)	pág. 17
PAGO(tasa, nper, va, vf, tipo)	pág. 16
PAGO.INT.ENTRE(tasa, nper, vp, per_inicial, per_final, tipo)	pág. 20
PAGO.PRINC.ENTRE(tasa, nper, vp, per_inicial, per_final, tipo)	pág. 21
PAGOINT(tasa, período, nper, va, vf, tipo)	pág. 18
PAGOPRIN(tasa, período, nper, va, vf, tipo)	pág. 20
PRECIO(liq, vencido, tasa, rendto, valor_de_rescate, frec, base)	pág. 39
PRECIO.DESCUENTO(liq, vencido, descuento, valor_de_rescate, base)	pág. 31
PRECIO.VENCIMIENTO(liq, vencido, emisión, tasa, rendto, base)	pág. 37
RENDTO(liq, vencido, tasa, precio, valor_de_rescate, frec, base)	pág. 40
RENDTO.DESC(liq, vencido, precio, valor_de_rescate, base)	pág. 32
RENDTO.VENCTO(liq, vencido, emisión, tasa, precio, base)	pág. 39
TASA(nper, pago, va, vf, tipo, estimar)	pág. 17
TASA.DESC(liquidación, vencido, precio, valor_de_rescate, base)	pág. 34
TASA.INT(liq, vencido, inversión, valor_de_rescate, base)	pág. 30
TASA.NOMINAL(tasa_efectiva, núm_per)	pág. 9
TIR(valores, estimar)	pág. 25
TIR.NO.PER(valores, fechas, estimar)	pág. 28
TIRM(valores, tasa_financiamiento, tasa_reinversión)	pág. 26
VA(tasa, nper, pago, vf, tipo)	pág. 15

VF(tasa, nper, pago, va, tipo)	pág. 15
VF.PLAN(capital, plan_serie_de_tasas)	pág. 37
VNA(tasa, valor1, valor2, ...)	pág. 23
VNA.NO.PER(tasa, valores, fechas)	pág. 27